



new location is special in some way. This is also the best place to incorporate the following piece of code to decide whether the program should generate random spooks:

```
2707 REM ** RANDOM GHOST **
2710 IF P>4 AND RND(1)<0.1 THEN GOSUB 4290:RETURN
```

Line 2710 first of all ensures that the current location has not already been designated as special, since ghosts appearing in the middle of special routines could make life very complex. If the location is ordinary then, using the RND command, there is a 1-in-10 chance that the program will produce a ghost. RND commands generate 'pseudo-random' numbers — so called because the pattern of numbers generated from power-up is predictable. To make the sequence less predictable, we use the RND command with a negative operand in the case of the BBC Micro and Commodore 64, and the RANDOMISE command for the Spectrum (see 'Basic Flavours').

```
287 R=RND(-1)
```

If the 'ghosts' routine is called, then we enter another special scenario in which the player is confronted by the ghostly apparition. The routine follows the usual procedure: it generates an initial message, asks for an instruction and splits the instruction into the verb and the rest of the sentence. Normal commands are dealt with by the standard subroutine, but, again, the GO command is suppressed — a message informs the player that, being transfixed with terror, he cannot move.

SAFETY NET

New commands can be dealt with at this stage. In common with the other special location handling routines, the quality of the finished game depends upon how much programming effort is put into designing these routines. Any command not directly useful in the routine can be dealt with by the I DON'T UNDERSTAND safety net. With additional programming effort, however, we can handle commands that might be expected of the player, but which will not help the situation. An example of this approach is used in the 'ghosts' routine.

If a player is confronted with a ghost, his first thought might be to FIGHT or KILL the ghost (if you can kill ghosts!). The 'ghosts' routine deals with these two commands by calling a special subroutine. This subroutine simply displays a message that these instructions do not assist the player, but does so in a way that is substantially more attractive than simply reporting I DON'T UNDERSTAND.

```
4290 REM **** RANDOM GHOST S/R ****
4295 SE=1:BC=0
4300 SN#="YOU FEEL A COLD SENSATION RUNNING THE LENGTH"
4305 SN#="OF YOUR SPINE. SUDDENLY A WHITE APPARITION"
4310 SN#="APPEARS FROM OUT OF THE TREES AND"
4315 SN#="MOVES TOWARDS YOU":GOSUB5500:REM FORMAT
4320 :
```

```
4325 SN#="THE GHOST MOVES CLOSER":GOSUB5500
4330 GC=GC+1:IF GC>4 THEN GOSUB4455:REM
4335 PRINT:INPUT"INSTRUCTIONS":IS#
4340 GOSUB2500:REM SPLIT INSTRUCTION
4345 IF F=0 THEN 4325:REM NEXT INSTRUCTION
4350 OP=P:GOSUB3000:REM ANALYSE INSTRUCTION
4355 IF MF=1 AND VB#="GO" THEN GOSUB4400:GOTO 4325
4357 IF MF=1 AND VB#="LOOK" THEN GOSUB2000:GOSUB2300:GOTO4325
4360 IF VF=1 THEN 4325:REM NEXT INSTRUCTION
4365 REM ** NEW INSTRUCTION WORDS **
4370 IF VB#="KILL" OR VB#="FIGHT" THEN GOSUB4425:GOTO 4325
4375 :
4385 IF VB#="SING" THEN GOSUB4500:RETURN
4390 SN#="I DON'T UNDERSTAND":GOSUB5500:GOTO4325
4395 :
4400 REM ** ATTEMPT TO MOVE **
4405 SN#="YOU ARE TRANSFIXED WITH TERROR AND CANNOT"
4410 SN#="MOVE...YET":MF=0:GOSUB5500:P=OP
4415 RETURN
4420 :
4425 REM ** FIGHT OR KILL **
4430 SN#="THE GHOST IS A BEING OF THE SUPERNATURAL"
4435 SN#="AND LAUGHS AT YOUR FEEBLE ATTEMPTS"
4440 SN#="TO INJURE HIM":GOSUB5500
4445 RETURN
4450 :
4455 REM ** DEATH **
4460 SN#="THE PAIN IN YOUR CHEST BECOMES UNBEARABLE"
4465 SN#="AND YOU SLUMP ONTO THE LEAFY FOREST FLOOR.":GOSUB5500
4470 SN#="YOUR SPIRIT RISES FROM YOUR INERT BODY"
4475 SN#="AND YOU FLOAT AWAY INTO THE MIST TO JOIN"
4480 SN#="THE OTHER TORMENTED SOULS OF THE"
4485 SN#="HAUNTED FOREST.":GOSUB5500
4490 END
```

STING IN THE TAIL

If any of the normal commands, or commands that are of no use to the player, are issued, the routine will obey them if possible and loop back for the next instruction. There is a sting in the tail of this routine, because a count is kept of the number of instructions issued by the player while being confronted by the ghost. If more than four instructions are issued, then the ghost moves in to kill the player. The only way that the player can escape is to SING a song. If the player elects to sing, he is given a choice of three songs, one of which (randomly chosen) will appease the ghost. If, however, the wrong tune is chosen, the player's spirit will join the army of tormented souls who have lost their way in the Haunted Forest:

```
4500 REM ** SING **
4505 SN#="YOU KNOW THREE SONGS. WHICH ONE WILL YOU CHOOSE?":GOSUB5500
4510 SN#="1) THE THEME FROM 'GHOSTBUSTERS':":GOSUB5500
4515 SN#="2) 'THERE'S A GHOST IN MY HOUSE':":GOSUB5500
4520 SN#="3) 'WAY DOWN UPON THE SWANEE RIVER':":GOSUB5500
4525 PRINT:INPUT"MAKE YOUR CHOICE":C#
4530 IF VAL(C#)>3 OR VAL(C#)<1 THEN PRINT:PRINT"INVALID":GOTO4525
4535 CR=INT(RND(1)*3)+1
4537 IF CR<>VAL(C#) THEN GOSUB4542:REM WRONG TUNE
4540 GOSUB4565:REM CORRECT
4542 REM **** WRONG TUNE S/R ****
4545 SN#="THE GHOST HAS A PARTICULAR HATRED OF"
4550 SN#="THAT TUNE AND LUNGES AT YOU.":GOSUB5500
4555 GOSUB 4455:REM DEATH
4560 :
4565 REM ** CORRECT TUNE **
4570 SN#="THE GHOST IS APPEASED BY YOUR RENDITION OF THE TUNE"
4575 SN#="AND VAPOURISES INTO THIN AIR":GOSUB5500
4580 RETURN
```